Formulating and Solving Exhaustible Resource Models as Mixed Complementarity Problems in GAMS

Sven M. Flakowski*
Westfälische wilhelms-Universität Münster
Institute of Economic Theory

Abstract

In this paper a formulation of dynamic non-linear programs as mixed complementarity problems (mcp) is shown. Models of exhaustible resource markets are used to describe the transformation. Once the mixed complementarity formulation is developed, the implementation in GAMS is described in detail.

Introduction

Modelling languages such as GAMS[1] (General Algebraic Modelling System) and AMPL[2] (A Modelling Language for Mathematical Programming) are becoming more and more important in applied economics.[3] As a result of the ease with which they can represent and solve mathematical expressions, they have become a very powerful tool of economic modellers. The handling of a great amount of data is made comfortable, and therefore economic modellers are in a position to concentrate on the model itself instead of developing suitable solution algorithms.[4]

By using mixed complementarity problem (mcp) formulations, many special cases of mathematical problems can be solved, such as linear and non-linear equations, linear and non-linear complementarity problems, and linear and non-linear programs, where the latter is used to derive the optimal price and extraction path of a non-renewable natural resource. The main advantage of an mcp formulation lies in its flexibility and speed solving complex economic models, which results from the fact that first-order conditions are used to set up the models. With the excellent exceptions of Rutherford (1995) and Ferris/Munson (2000) on static examples of mcp, literature on formulating economic problems as non-linear complementarity problems is rare. Usually, the presentation of complementarity problems is done without describing the economic content of the underlying problem.[5]

The goal of this paper is to give an overview of the formulation of complementarity problems and to show their concrete implementation in GAMS by modelling well-known dynamic problems of non-renewable resources theory. The paper addresses economists with a basic training in GAMS and control theory, who are interested in making use of the powerful tool of mcp algorithms. To alleviate access to the problem, the next section describes and explains the general formulation of complementarity problems and ends with an illustration of a non-linear program as an mcp. The third section recalls the basic theory of non-renewable resources on the assumption of three different market structures: perfect competition, monopoly and Nash–Cournot oligopoly. The implementation of mixed complementarity problems in GAMS is shown in the fourth section.

Mixed complementarity problems

The general mathematical formulation of mixed complementarity problems is as follows:

Given a function $F : R^N \rightarrow R^N$ and the bounds $l, u \in R^N$,

find $z, w, v \in R^N$

that meet the following conditions:

$$F(z) = w - v \qquad (0.1)$$

$$l \leq z \leq u,\ w \geq 0,\ v \geq 0 \qquad (0.2)$$

$$(z - l)^T \cdot w = 0 \qquad (0.3)$$

$$(u - z)^T \cdot v = 0 \qquad (0.4)$$

$$\text{with } -\infty \leq l \leq z \leq u \leq \infty \qquad (0.5)$$

The superscript T refers to the transposition of a matrix or a vector. To illustrate the formulation above, it makes sense to recall the derivation of the Kuhn–Tucker conditions.[6]



Figure 1. Graphical depiction of Kuhn–Tucker conditions
Source: Chiang (1984), p. 723.

Consider a problem with non-negativity restrictions only:

$$\begin{array}{ll} \text{Maximise} & f(z) \\ \text{s.t.} & z \geq 0 \end{array} \qquad (0.6)$$

where the function f is assumed to be differentiable.

This is an optimization problem with one inequality constraint; the variable z is lower bounded at 0. Now there are three possibilities for a local maximum of the function $f(z)$ subject to $z \geq 0$. First, a local maximum can be an interior solution at point A in diagram (a) of Figure 1. The resulting first-order condition is well known: $f'(z) = 0$. Second, a local maximum can also occur at point B in diagram (b) at the lower bound of z, where $z = 0$. Even in this case the first-order condition $f'(z) = 0$ is still valid. Third, a local maximum is shown in diagram (c). Again the local maximum is at the lower bound $z = 0$ but in this case we obtain $f'(z) < 0$.

Thus one of the following conditions must hold to determine a local maximum of (0.6):

$$\begin{array}{lll} f'(z) = 0 & \text{and} & z > 0 \\ f'(z) = 0 & \text{and} & z = 0 \\ f'(z) < 0 & \text{and} & z = 0 \end{array} \qquad (0.7)$$

These conditions can be combined to the following statement:

$$f'(z) \leq 0, \qquad z \geq 0, \qquad z \cdot f'(z) = 0 \qquad (0.8)$$

What does condition (0.8) mean for the formulations in (0.1)–(0.4)? If z is at its lower bound l, i.e. $z = l$ (in (0.7) $z = 0$), $F(z) = 0$ or $F(z) < 0$ holds. It follows that either $w = 0$ and thus $F(z) = 0$ or $w > 0$ and thus $F(z) < 0$. If z is neither at its lower bound nor at its upper bound (in (0.7) $z > 0$), i.e. $z \neq l,u$, because of (0.3) and (0.4) $w,v = 0$ holds and consequently we obtain $F(z) = 0$ ($f'(z) = 0$ and $z > 0$ in (0.7)). Similar considerations hold for an upper bound on z and as a consequence, we obtain equations (0.1)–(0.4) that combine all possible combinations.

Now it is possible to state a non-linear program as a complementarity problem. Consider the following non-linear optimization problem:

$$\underset{x}{\text{minimize}}\; f(x)$$

s.t.:

$$x \in X := \{x \mid g(x) \leq 0,\; x \geq 0\}$$

with $x \in R^n$ and $f : R^n \to R^m$ as well as $g : R^n \to R^m$ representing continuous differentiable functions. The resulting necessary first-order conditions are:

$$\begin{array}{rll} \nabla f(x) + y \cdot \nabla g(x) \geq 0, & x \geq 0, & \perp \\ -g(x) \geq 0, & y \geq 0, & \perp \end{array} \qquad (0.9)$$

The sign $\perp$ means that the inner product of two matrices or vectors is 0 (they are orthogonal) so that the following condition holds: $x^T \cdot [\nabla f(x) + y \cdot \nabla g(x)] = 0$

Condition (0.9) can be restated in the mcp format of (0.1)–(0.4):

$$z = \begin{pmatrix} x \\ y \end{pmatrix},\; l = \begin{pmatrix} 0 \\ 0 \end{pmatrix},\; u = \begin{pmatrix} +\infty \\ +\infty \end{pmatrix},\; F(z) = \begin{cases} \nabla f(x) + y^T \cdot \nabla g(x) \\ -g(x) \end{cases} \qquad (0.10)$$

Now vector z has a lower bound at $l = 0$. As there is no upper bound on z, according to (0.4) $v = 0$ must be valid. If z is not at its lower bound ($z > 0$), $w = 0$ must also be valid. But if $z = 0$, w is not determined and can be larger than 0 as well as equal to 0. $w = 0$ is the representation of point B in Figure 1 and $w > 0$ the representation of point C.

In a literal sense, (0.10) is not a mixed complementarity problem, but a pure complementarity problem. If the lower bound of variable x is relaxed so that no explicit lower bound is valid, we get from (0.9):

$$\begin{array}{rll} \nabla f(x) + y \cdot \nabla g(x) = 0, & x \text{ free}, & \perp \\ -g(x) \geq 0, & y \geq 0, & \perp \end{array} \qquad (0.11)$$

The resulting mixed complementarity problem is as follows:

$$z = \begin{pmatrix} x \\ y \end{pmatrix},\; l = \begin{pmatrix} -\infty \\ 0 \end{pmatrix},\; u = \begin{pmatrix} +\infty \\ +\infty \end{pmatrix},\; F(z) = \begin{cases} \nabla f(x) + y^T \cdot \nabla g(x) \\ -g(x) \end{cases} \qquad (0.12)$$

In this case x is unbounded. From (0.3) and (0.4) we get $w,v > 0$ and therefore from (0.1) follows $F(x) = 0$ (point A in Figure 1). Further illustrations of special cases represented by mixed complementarity problems can be found in Rutherford (1995), Dirkse and Ferris (1994) and Ferris and Munson (2000).

Models of exhaustible natural resource markets

The most simplified model of an exhaustible natural resource market describes a market for an exhaustible resource which is supplied by a large number of producers n.[7] This is the well-known perfect competition case. Producers compete with each other so that none of them is able to influence the market price p_t by a variation of his production quantities. At time $t = 0$ each producer owns a given and known resource stock with $S^i(0) = S^i_0$ where $i = 1,...,n$ representing the number of producers. Let R^i_t be the quantities extracted by producer i at time t and let there be identical but constant marginal extraction costs $dC(R_t)/dR_t = c \geq 0$ so that there are no stock-related extraction costs. The producers are faced with a linear demand function of type

$$D_t = a - b \cdot p_t \tag{0.13}$$

whereas D_t represents demand for the resource at time t. In equilibrium the following condition has to hold:

$$D_t = R_t = \sum_i R^{it} \qquad \forall t \tag{0.14}$$

With a positive discount rate r the maximization problem for a single producer can be stated as follows:[8]

$$\max_{R^i_t} \int_0^\infty \left[(p_t - c) \cdot R^i_t \right] \cdot e^{-r \cdot t} \, dt \tag{0.15}$$

s.t.

$$\dot{S}^i_t = R^i_t \tag{0.16}$$

$$S^i_t \geq 0 \tag{0.17}$$

$$R^i_t \geq 0 \tag{0.18}$$

and the initial condition:

$$S^i = S^i_0 \tag{0.19}$$

The subscript i will be ignored in the following considerations, because all producers are assumed to be identical and only the industry outcome is analysed. To calculate the individual extraction paths, total quantity at time t has to be divided by the number of producers n.

The present value Hamiltonian for the maximization problem (0.15)–(0.18) is:

$$H_t = (p_t - c) \cdot R_t - \lambda_t \cdot R_t \tag{0.20}$$

The resulting first-order conditions for a maximum are:

$$\frac{\partial H_t}{\partial R_t} = p_t - c - \lambda_t \leq 0, \qquad R_t \geq 0, \qquad R_t \cdot (p_t - c - \lambda_t) = 0 \tag{0.21}$$

$$\dot{S}_t = \frac{\partial H_t}{\partial \lambda_t} = -R_t \tag{0.22}$$

$$\frac{\partial H_t}{\partial S_t} = r \cdot \lambda_t - \dot{\lambda}_t \geq 0, \qquad S_t \geq 0, \qquad \perp \tag{0.23}$$

Solving (0.13) for p_t and substituting in (0.21) leads under consideration of (0.14) to

$$\lambda_t \geq \frac{(a - R_t)}{b} - c, \qquad R_t \geq 0, \qquad \perp \tag{0.24}$$

Equations (0.24), (0.22) and (0.23) are the first-order conditions for the optimization problem (0.15)–(0.19). It can be restated as a mixed complementarity problem of the form (0.1)–(0.4):

$$z = \begin{pmatrix} R \\ S \\ l \end{pmatrix}, \quad l = \begin{pmatrix} 0 \\ 0 \\ -\infty \end{pmatrix}, \quad u = \begin{pmatrix} \infty \\ \infty \\ \infty \end{pmatrix}, \quad F(z) = \begin{cases} p_t - c - \lambda_t \\ \dot{\lambda}_t - r \cdot \lambda_t \\ \dot{S}_t + R_t \end{cases} \tag{0.25}$$

The second model to be analysed is the monopoly case with $n = 1$. In contrast to the case of the competitively organised market, the monopolist has an influence on the market price through a variation of the quantities extracted. Equation (0.15) changes into:

$$\max_{R_t} \int_0^\infty \left[(p_t(R_t) - c) \cdot R_t \right] \cdot e^{-r \cdot t} \, dt \tag{0.26}$$

Now the market price depends on the extraction quantities of the monopolist. The demand function (0.13) can be directly substituted in (0.26), so that the resulting Hamiltonian looks as follows:

$$H_t = \left[\frac{(a - R_t)}{b} - c \right] \cdot R_t - \lambda_t \cdot R_t \tag{0.27}$$

The first-order conditions (0.22) and (0.23) remain unchanged, but (0.24) has to be altered:

$$\lambda_t \geq \frac{(a - 2 \cdot R_t)}{b} - c, \qquad R_t \geq 0, \qquad \perp \tag{0.28}$$

This completes the first-order conditions of the monopoly case.

Now consider an oligopolistic market with only a few (minimum two) producers (players). Each player is aware not only of his own influence on the market price, but of the fact that the quantities of all other players have an influence, too. An open-loop Nash–Cournot equilibrium is reached if all players take the quantities of all the other players as given and optimize their own extraction paths with respect to these quantities.[9] When an open-loop Nash–Cournot equilibrium is reached, no player has an incentive to deviate from the chosen path, because all players give reciprocally best answers to the other's strategies. An open-loop strategy characterises a single-staged decision process, i.e. each player chooses all actions at the beginning of the game independently of the actual evolution of the game. The players make binding agreements at the beginning of the game over the actions during the game. This may cause problems concerning the time consistency of the game, because a strategy chosen at the beginning of the game may prove itself suboptimal during the game. As a result of this, the players may have incentives to deviate from their original strategies.[10]

Consider a single producer i at time $t = 0$ holding a given and known stock of resource $S^i(0) = S^i_0$ with $i = 1,...,n$. Extracted quantities of player i at time t are R^i_t and the linear demand function (0.13), as well as equation (0.14), is valid.

Extraction of the resource incurs individual constant marginal extraction costs c^i. The optimization problem can be stated as follows:

$$\max_{R_t^i} \int_0^\infty \left[\left(p_t(R_t) - c^i \right) \cdot R_t^i \right] \cdot e^{-r^i \cdot t} \; dt \tag{0.29}$$

s.t.

$$\dot{S}_t^i = -R_t^i \tag{0.30}$$

$$S_t^i \geq 0 \tag{0.31}$$

$$R_t^i \geq 0 \tag{0.32}$$

Substituting (0.13) in (0.29) with regard to (0.14) yields

$$\max_{R_t^i} \int_0^\infty \left[\left(\frac{a - R_t}{b} - c^i \right) \cdot R_t^i \right] \cdot e^{-r^i \cdot t} \; dt \tag{0.33}$$

The resulting present value Hamiltonian is:

$$H_t^i = \left(\frac{a - R_t}{b} - c^i \right) \cdot R_t^i - \lambda_t^i \cdot R_t^i \tag{0.34}$$

and the first-order conditions for a maximum are:

$$\frac{\partial H_t^i}{\partial R_t^i} = \frac{a - R_t - R_t^i}{b} - c^i - \lambda_t^i \leq 0, \qquad R_t^i \geq 0, \qquad \perp \tag{0.35}$$

$$\dot{S}_t^i = \frac{\partial H_t^i}{\partial \lambda_t^i} = -R_t^i \tag{0.36}$$

$$\frac{\partial H_t^i}{\partial S_t^i} = r^i \cdot \lambda_t^i - \dot{\lambda}_t^i \geq 0, \qquad S_t^i \geq 0, \qquad \perp \tag{0.37}$$

In contrast to (0.28), condition (0.35) is now not only dependent on the quantities of player i but in addition on the quantities of all players R_t. Figure 2 recapitulates the first-order conditions of the three market structures considered in this chapter.

The first-order conditions are necessary to derive optimal price and extraction paths by using the mixed complementarity formulation. They correspond to the function $f(z)$ in equations (0.1), (0.10) and (0.12) respectively. But now vector z comprises the variables R, S, and λ. The next section shows the implementation in GAMS.

Implementation in GAMS

The GAMS code for the mcp formulation of the competition case is shown in Figure 3.[11] First of all, the necessary sets, parameters and scalars are defined. It is important to notice that the price is defined not as a variable, but as a parameter. This is because the first-order conditions (0.22)–(0.24) do not comprise the price explicitly. Therefore the price path has to be calculated separately.[12] Subsequently the definition of variables follows: `R(t)` and `S(t)` are positively defined (or lower bounded), i.e. `R(t),S(t)≥0; l(t)` is defined as an unbounded variable. The functions defined by `Prof(t)..`, `MoveS(t)..` and `MoveL(t)..` reproduce the first-order conditions of (0.22)–(0.24). The '`Model`' statement assigns equations to their corresponding complementarity variables. Note that in GAMS the sign is replaced by a `'.'`. Thus the function defined by `Prof(t)..` is complementary to the variable `R(t)`, whereas the function defined by `Move(t)` is complementary to the variable `S(t)`. It is important to notice that only positive defined variables can be made complementary to inequalities or equations, whereas free variables can only be made complementary to equations. But it is not necessary to match free variables to equations; it is only required to have the same number of equations as free variables.[13]

Mathematically the following conditions have to hold:[14]

$$\begin{array}{lll} \lambda_t - \dfrac{(a - R_t)}{b} + c \geq 0, & R_t \geq 0, & \perp, \forall t \\ \lambda_t \cdot r - (\lambda_{t+1} - \lambda_t) \geq 0, & S_t \geq 0, & \perp, \forall t \\ S_t - S_{t-1} + R_t = 0, & \lambda_t \text{ free}, & \perp, \forall t \end{array} \tag{0.38}$$

The function defined by `Prof(t)..` specifies that the shadow price `l(t)` is no smaller than the difference between marginal revenue (the price in the perfectly competition case) and marginal costs. Equation `MoveL(t)..` specifies how the shadow price can change over time. Finally, equation `Move(t)..` specifies how the extraction quantity `R(t)` relates to the stock of the resource `S(t)`. Note that the assignment of the starting value of the resource stock cannot be done by fixing the value of `S(0)`. By using

```
t0(t) = yes$(ord(t) eq 1),
```

Perfect Competition	Monopoly	Nash-Cournot-Oligopoly
$\lambda_t \geq \dfrac{(a - R_t)}{b} - c, \; R_t \geq 0, \; \perp$	$\lambda_t \geq \dfrac{(a - 2 \cdot R_t)}{b} - c, \; R_t \geq 0, \; \perp$	$\lambda_t^i \geq \dfrac{a - R_t - R_t^i}{b} - c^i, \; R_t^i \geq 0, \; \perp$
$\dot{S}_t = -R_t$	$\dot{S}_t = -R_t$	$\dot{S}_t^i = -R_t^i$
$\dfrac{\partial H_t}{\partial S_t} = r \cdot \lambda_t - \dot{\lambda}_t \geq 0, \; S_t \geq 0, \; \perp$	$\dfrac{\partial H_t}{\partial S_t} = r \cdot \lambda_t - \dot{\lambda}_t \geq 0, \; S_t \geq 0, \; \perp$	$\dfrac{\partial H_t^i}{\partial S_t^i} = r^i \cdot \lambda_t^i - \dot{\lambda}_t^i \geq 0, \; S_t^i \geq 0, \; \perp$

Figure 2. Market structures and first-order conditions

```
Sets
  t                     time period                  /1*50/
  t0(t)                 time period subset;

Parameter
price(t)                price in period t;

Scalars
  a                     demand at zero price         /80/
  b                     absolute slope of demand     /0.2/
  c                     marginal extraction costs    /3/
  S0                    initial stock at t0          /500/
  z                     discount rate                /0.05/;

Variable
  l(t)                  shadow price at time t;

Positive Variables
  R(t)                  extracted quantities at time t
  S(t)                  stock at time t;

Equations
  Prof(t)               profit condition
  MoveS(t)              movement of stock
  MoveL(t)              movement of shadow price;

Prof(t)..               l(t)+c =g= (a - R(t))/b;
MoveL(t)..              l(t) + z*l(t) =g= l(t+1);
MoveS(t)..              S0$t0(t) + S(t-1) -R(t) =e= S(t);

t0(t) = yes$(ord(t) eq 1);

Model competition /prof.r, movel.s, moves/;

Solve competition Using MCP;

price(t) = (a - R.l(t))/b;

Display price, r.l, s.l, l.l;
```

Figure 3. Competitively organised market as an mcp in GAMS

the logical value 'yes' is assigned to the label '1' of set `t0` (thus the value 'yes' is assigned to time period '1'). In the function

```
MoveS(t)..  S0$t0(t) + S(t-1) -R(t) =e= S(t);
```

the scalar `S0` (the initial stock) is added if the condition `t0(t)` is true, i.e. the logical value of `t0` is 'yes'. Using this procedure makes sure that the scalar representing the initial stock `S0` is only added in the first period. In all other periods only the equation S(t–1) –R(t) = e = S(t) has to hold.

The only difference between the perfect competition case and the monopoly case is in the first-order condition:

$$
\begin{aligned}
&\lambda_t - \frac{(a-2\cdot R_t)}{b} + c \geq 0, && R_t \geq 0, && \perp, \forall t \\
&\lambda_t \cdot r - (\lambda_{t+1} - \lambda_t) \geq 0, && S_t \geq 0, && \perp, \forall t \\
&S_t - S_{t-1} + R_t = 0, && \lambda_t \text{ free}, && \perp, \forall t
\end{aligned}
\tag{0.39}
$$

Compared to Figure 3 in the GAMS code only the function defined by `Prof(t)..` has to be changed:

```
Prof(t)..          l(t)+ c =g= (a – 2*R(t))/b;
```

Now the marginal revenue of the monopolist is no longer identical to the demand function. The monopolist takes into account his influence on market price and it is well known that the slope of the marginal revenue function is twice that of the inverse demand function.

Somewhat more elaborate than the monopoly case is the formulation of the Nash–Cournot equilibrium. The complete GAMS code is shown in Figure 4. An additional set `i` representing the individual players is introduced (in this case `player 1` and `player 2`) and consequently initial stocks, marginal extraction costs and discount rates have to be defined individually for each player. Variables `R(i,t), S(i,t)` and `l(i,t)` are also defined over the set `i`. The Nash–Cournot case as an mcp is defined as follows:

```
Sets
  t                    time period                     /1*50/
  t0(t)                time period subset
  i                    players                         /player1, player2/;
Alias (i,j);

Parameter
  S0(i)                initial stocks
                                                       /player1 500
                                                        player2 500/

  c(i)                 marginal extraction costs
                                                       /player1 3
                                                        player2 5/

  z(i)                 discount rate
                                                       /player1 0.05
                                                        player2 0.05/

Scalars
  a                    demand at zero price            /80/
  b                    absolute slope of demand        /0.2/

Variables
  l(i,t)               shadow prices
  p(t)                                                 price in t

Positive Variables
  R(i,t)               extracted quantities of player i at t
  S(i,t)               stock of resource of player i at t

Equations
  Prof(i,t)            profit condition
  MoveL(i,t)           movement of shadow prices
  MoveS(i,t)           movement of stocks
  Price(t)             demand function;

Prof(i,t)..            l(i,t)+c(i) =g= (a-sum(j,R(j,t))-R(i,t)) /b;
MoveL(i,t)..           l(i,t) + z(i)*l(i,t) =g= l(i,t+1) ;
MoveS(i,t)..           S0(i)$t0(t) + S(i,t-1) - R(i,t) =e= S(i,t) ;Price(t)..    p(t) =e= (a
- sum(i, R(i,t)))/b;

Model Oligopoly /prof.r, MoveL.s, MoveS, Price/;
t0(t) = yes$(ord(t) eq 1);Solve Oligopoly Using MCP;
Display r.l, s.l, l.l, p.l;
```

Figure 4. Two-player Nash–Cournot equilibrium as an MCP in GAMS

$$
\begin{array}{lll}
\lambda_t^i - \dfrac{a - R_t - R_t^i}{b} + c^i \geq 0, & R_t^i \geq 0, & \perp, \forall i,t \\
\lambda_t^i \cdot r^i - (\lambda_{t+1}^i - \lambda_t^i) \geq 0, & S_t^i \geq 0, & \perp, \forall i,t \\
S_t^i - S_{t-1}^i + R_t^i = 0, & \lambda_t^i \text{ free}, & \perp, \forall i,t
\end{array}
\qquad (0.40)
$$

The individual shadow prices `l(i,t)` for each player are defined in equation `Prof(i,t)..` and are not only dependent on the individual extraction quantities of player `i`, but on the extracted quantities of all players `sum(j,R(j,t))`. The `set j` is assigned to the `set i` by using the `Alias` statement. Now the close connection between the three different market structures can be shown. If only one player is defined in the `set i`, the term `sum(j,R(j,t))` changes into `R(j,t)` so that the function defined by `Prof(i,t)..` changes into:

```
Prof(i,t)..     l(i,t)+c(i) =g= (a - R(j,t) -
                                     R(i,t)) /b;
```

Because `sets i` and `j` are identical, we have the results of the monopoly case.

Now consider the assumption of many identical players in the perfect competition case. The quantities extracted by a single player `R(i,t)` in relation to the sum of the quantities extracted by all players `sum(j,R(j,t))` is very small, i.e. the quantities of an individual player do not have any measurable influence on the shadow price. The function defined as

```
Prof(i,t).. l(i,t)+c(i) =g= (a-sum(j,R(j,t))-
                                     R(i,t)) /b;
```

can then be restated as

```
Prof(i,t)..l(i,t)+c(i) = g = (a - sum(j,R(j,t)) )
                                              /b;
```

without changing the outcome. Because of (0.14), this is the perfect competition case as stated in Figure 3.

The following function is used to determine the market price:

```
Price(t).. p(t) = e = (a - sum(i, R(i,t)))/b;
```

The variable `p(t)` is unbounded and therefore must not be matched to an equation in the model statement. Thus the market price is determined residually from the sum of the extracted quantities by using the demand function (0.13).[15]

Note that in contrast to the non-linear program, the results of the model cannot be checked by observing the development of marginal rents with respect to time. This development is explicitly specified in the mcp. As a consequence of this, the solution will always follow this rule, irrespective of errors in the first-order conditions. An error in the derivatives of the Hamiltonians, for example, can lead to an unobserved error in the results if GAMS is able to compute a feasible solution, because marginal rent will nevertheless grow with the discount factor. Therefore the determination of first-order conditions and their implementation as mcp in GAMS has to be done with great care.

Summary

This paper has shown an alternative method to formulate dynamic non-linear programs. Mixed complementarity problems have been presented and the formulation of non-linear programs as mixed complementarity problems has been described. By means of three different models of exhaustible resource markets (perfect competition, monopoly and Nash–Cournot equilibrium) the transformation of non-linear programs as complementarity problems has been shown and their implementation in GAMS described. Once the mixed complementarity formulation is developed the implementation in GAMS has proved rather straightforward. It has been shown that the complementarity formulation is a powerful tool for implementing and solving models of exhaustible natural resource markets in GAMS.

Notes

* The helpful comments and suggestions of Eric Meyer, Alexander Smajgl and Valerie Böhner are gratefully acknowledged.

1 A lot of information about GAMS can be found at www.gams.com.

2 See the extensive information at www.ampl.com.

3 At the beginning of the 1980s, GAMS was originally developed to support economists at the World Bank in the quantitative analysis of policy scenarios. Linear programming was the standard approach to economic modelling at that time and, not surprisingly, solution algorithms for mainly this type of mathematical problems were developed. As in the following years the spectrum of available solution methods broadened, lots of (not only) economic problems became solvable – especially non-linear (nlp), integer (ip), mixed integer (mip) and mixed integer non-linear problems (minlp). It was not until the middle of the 1990s that solution algorithms were developed for complementarity problems (cp) and mixed complementarity (mcp) problems respectively. Mixed complementarity problems are sometimes also called 'generalised complementarity problems'. See in detail Billups *et al.* (1997) and Dirkse and Ferris (1995).

4 See Ferris and Munson (2000).

5 See, for example, Ferris and Sinapiromsaran (2000) and Billups and Murty (1999).

6 See Kuhn and Tucker (1951). An extensive and excellent presentation can be found in Chiang (1984), S. 722 ff.

7 Markets of exhaustible resources have been analysed by a lot of authors; see, for example, Hotelling (1931), Hoel (1978), Dasgupta and Heal (1979), Hartwick (1989) and Conrad (1999).

8 Although the model is subsequently implemented in GAMS by using a discrete-time approach, the underlying economic basics are presented in a continuous time format in this section. This is done as the continuous time formulation is much more common in economic literature and the discrete time formulation would not lead to any different results in this rather simple context.

9 The open-loop Nash–Cournot equilibrium on markets for exhaustible resources is extensively discussed in economic literature; see, for example, Salant (1976), Ulph and Folie (1980), and Lewis and Schmalensee (1980).

10 The problem of time inconsistency is extensively discussed in Newbery (1981).

11 The PATH solver is used to calculate the model results.

12 There is no need to calculate the price by using a parameter as it is done in Figure 3. It can also be calculated by using a variable, but this variable would not be part of the underlying mcp. Calculating the price by a variable is shown in the Nash–Cournot case in Figure 4.

13 See Ferris and Munson (2000), p. 10.

14 In discrete-time formulation.

15 In contrast to Figure 3 the market price is calculated by an equation and not by an assignment of a parameter after the model is solved. It is self-evident that the model becomes (much) larger, but in this case the formulation is presented as an alternative solution method for academic purposes only.

References

Billups, S. C. and Murty, K.G. (1999) *Complementarity Problems*, Center for Computational Mathematics Series of Reports, No. 147.

Billups, S. C., Dirkse, S. P. and Ferris, M. C. (1997) 'A comparison of large scale mixed complementarity problem solvers, *Computational Optimization and Applications,* 7, pp. 3–25.

Chiang, A. C. (1984) *Fundamental Methods of Mathematical Economics* (3rd. edn), Auckland: McGraw Hill.

Conrad, J. M. (1999) *Resource Economics,* Cambridge.

Dasgupta, P. S. and Heal, G. M. (1979) *Economic Theory and Exhaustible Resources,* Cambridge: Cambridge University.

Dirkse, S. P. and Ferris, M. C. (1994) *MCPLIB – A Collection of Nonlinear Mixed Complementarity Problems,* University of Wisconsin, Computer Science Department.

Dirkse, S. P. and Ferris, M. C. (1995) 'The PATH Solver – a non-monotone stabilization scheme for mixed complementarity problems', *Optimization Methods and Software,* 5, 123–56.

Ferris, M. C. and Munson, T. S. (2000) 'Complementarity problems in GAMS and the PATH Solver', *Journal of Economic Dynamics and Control,* 24, pp. 165–88.

Ferris, M. C. and Sinapiromsaran, K. G. (2000) 'Formulating and solving nonlinear programs as mixed complementarity problems', in V. H. Nguyen, J. J. Strodiot and P. Tossings (eds), *Optimization*, Berlin, pp. 132–48.

Hartwick, J. M. (1989) *Non-renewable Resources Extraction Programs and Markets*, London: Routledge.

Hoel, M. (1978) *Resource Extraction under some Alternative Market Structures*, Meisenheim am Glan: Verlag Anton Hain.

Hotelling, H. (1931) 'The economics of exhaustible resources', *Journal of Political Economy,* 39, pp. 137–75.

Kuhn, H. W. and Tucker, A. W. (1951) 'Nonlinear programming, in J. Neyman (ed.), *Proceedings of the Second Berkeley Symposium on Mathematical Statistics and Probability,* Berkeley, 481–92.

Lewis, T. R. and Schmalensee, R. (1980) 'On oligopolistic markets for nonrenewable natural resources, *Quarterly Journal of Economics,* 95, pp. 475–91.

Newbery, D. M. (1981) 'Oil prices, cartels, and the problem of dynamic inconsistency', *The Economic Journal,* 91, pp. 617–46.

Rutherford, T. F. (1995) 'Extensions of GAMS for complementarity problems arising in applied economic analysis', *Journal of Economic Dynamics and Control,* 19, pp. 1299–324.

Salant, S. W. (1976) 'Exhaustible resources and industrial structure – a Nash–Cournot approach to the world oil market', *Journal of Political Economy,* 84, pp. 1079–93.

Ulph, A. M. and Folie, G. M. (1980) 'Exhaustible resources and cartels – an intertemporal Nash–Cournot model', *Canadian Journal of Economics,* 13, pp. 645–58.

Contact details

Sven M. Flakowski*
Westfälische wilhelms-Universität Münster
Institute of Economic Theory
Muenster 2003
Institute of Economic Theory
Universitaetsstr. 14–16, D-48143 Muenster

Tel: +49 (0)251-8322872, Fax: +49 (0)251-8328317
Email: 15svfl@wiwi.uni-muenster.de